ED 167 369 SE 026 736

TITLE A Computer Simulation of the U.S. Energy Crisis, Energy. Teacher Guide. Computer Technology Program Environmental Education Units.
INSTITUTION Northwest Regional Educational Lab., Portland, Oreg.
SPONS AGENCY National Inst. of Education (DHEW), Washington, D.C.
PUB DATE Sep 75
NOTE 28p.; For related documents, see SE 026 732-741; Contains occasional light and broken type
AVAILABLE FROM Office of Marketing, Northwest Regional Educational Lab., 710 S.W. Second Ave., Portland, Oregon 97204 ($3.25)

EDRS PRICE MF-$0.83 HC-$2.06 Plus Postage.
DESCRIPTORS Annotated Bibliographies; *Computer Assisted Instruction; *Energy; *Environmental Education; Higher Education; *Instructional Materials; Pollution; Secondary Education; *Simulation; Social Studies
IDENTIFIERS *Energy Education

ABSTRACT

This is the teacher's guide to accompany the student guide which together comprise one of five computer-oriented environmental/energy education units. The computer program, ENERGY, at the base of this unit, simulates the pattern of energy consumption in the United States. The total energy demand is determined by energy use in the various sectors such as the industrial sector, the transportation sector, the utilities sector, and so on. The demand for energy is shown to grow exponentially in each sector. Students are asked to balance supply of energy with demand by adjusting factors in each sector. This teacher's guide presents: (1) suggestions on introducing the unit; (2) student guide exercises and answers; (3) follow-up activities; and (4) an annotated source list. This unit is appropriate for social studies and environmental education courses grades 9 through 14. (MR)

COMPUTER TECHNOLOGY PROGRAM
ENVIRONMENTAL EDUCATION UNITS

A Computer Simulation of the U.S. Energy Crisis

```
EEEEE   N   N   EEEEE   RRRRR   GGGGG   Y   Y
E       NN  N   E       R   R   G   G   Y   Y
E       N N N   E       R   R   G        Y Y
EEEE    N N N   EEEE    RRRRR   G         Y
E       N N N   E       R R     G GGG     Y
E       N  NN   E       R  R    G   G     Y
EEEEE   N   N   EEEEE   R   R   GGGGG     Y
```

TEACHER GUIDE

NORTHWEST REGIONAL EDUCATIONAL LABORATORY
710 S.W. Second Avenue, Portland, Oregon 97204
Telephone (503) 248-6800

This work is submitted in confidence and has not as yet been published. The recipient receives the work in confidence and will use the same only for the purpose for which it is tendered.

This work represents development efforts to date and should not be considered final material. Therefore, use of this material is restricted exclusively to those persons and agencies cooperating in field testing.

Prospective users of this manual are urged to first run the sample simulation program provided in order to determine any needed or desirable adjustments prior to use.

September 1975

TABLE OF CONTENTS

INTRODUCTION TO THE UNIT

Unit Description

Subject Areas: Social Studies, Environmental Education

Topic: The U.S. Energy Crisis

Abstract: This unit is organized around an operating model which simulates important systemic aspects of the energy crisis in the United States. Working with this simulation, students will confront such questions as: "How can pollution be reduced without increasing the demand for energy?" and "If most of the population used mass transit instead of private cars, how much energy would be saved and how would this affect the quality of life?"

The computer program, ENERGY, at the base of the unit simulates the pattern of energy consumption in the United States. The total energy demand is determined by energy use in various sectors such as the industrial sector, the transportation sector, the utilities sector, and so on. The demand for energy is shown to grow exponentially in each of these sectors. The students' task will be to balance the supply of energy with the current demand by adjusting factors in each sector. Some of the adjustable factors are industry's production level, pollution standards and the availability of mass transit systems. In making these adjustments to reduce the demand for energy, students will be faced with the consequences of making these changes, as reflected in indices such as pollution level, economic well-being, general satisfaction level and so forth.

The ENERGY model is designed for use both as the basis for a role-playing game in which students represent different elements of society attempting to agree on solutions to the energy problem and as an experimental device for students to test various strategies and solutions.

Grade Level: 9-14

Computer Language: BASIC

Program Name: ENERGY

Objectives

The primary objectives of this unit are:

1. To provide an operational model of the energy system in the United States as a vehicle for the study of the energy problem.

2. To foster a better understanding of the energy crisis. Specifically, this understanding relates to:

   a. The causes of the energy crisis

   b. The nature of the growth of the energy shortage

   c. The pattern of energy use in the United States

   d. The relationship between the supply of energy and energy consumption

   e. The nature of the trade-offs that must be considered when changing the pattern of energy consumption

3. To provide students with a basic understanding of the complexity of the energy crisis. Reaching this objective will involve the students in comparing the model with their view of the problem.

Reasons for Studying the ENERGY Unit

In this age of crisis upon crisis, it is imperative that students be given an opportunity to confront and cope with real world problems. They need to engage in decision making and problem solving activities and exercises that involve setting priorities for energy use, acceptable sources of energy, the quality of the environment, economic changes and acceptable life styles. They need to develop the skills and insights necessary to understand the structure, functioning and impact of the various systems and subsystems that interact in complex ways to create a particular problem or crisis. Senator Howard H. Baker of Tennessee, a strong supporter of environmental education, states the need in the following way:

> Since the present problems result largely from ignorance and indifference, corrective action must alleviate these conditions. We cannot expect improvement unless public attitudes change, unless a better appreciation of environmental processes is inculcated in the present and future generations. Our great need today is a knowledgeable citizenry, conscious of their surroundings and willing to

> take the necessary social, economic and political steps to assure a better environment for all. The cost is great; but the benefits are still greater and the consequences of not acting will be profoundly tragic for ourselves and future generations.[1]

The Citizens Advisory Council on Environmental Quality in a recent report to the President pointed specifically to our education system charging that "our formal education system has done little to produce an informed citizenry sensitive to environmental problems, prepared and motivated to work toward their solution."[2]

Within recent years a renewed emphasis has been placed on environmental ecological education.[3] Educational materials for teachers and students concerning the concept of energy and its relationship to man and his environment, however, remain scarce.[4] Frank B. Brouillet, Superintendent of Public Instruction for the State of Washington, summarizes the need for energy related curriculum materials this way:

> As educators work to foster a greater understanding of society's environmental problems, we can best succeed by teaching the ecological relationships and principles that underlie problems such as Energy. We need to help students learn of Energy's relationship to man's environment, and learn to make decisions about possible alternative approaches and solutions to Energy problems.
>
> The Energy concerns of our society demand that the generation of students now in our schools become environmentally literate to the degree that they understand the environmental implications of human activities viewed from the perspective of social needs and values as they relate to general public policy.[5]

---

1. Baker, Howard H., Jr. "A New Direction for Federal Environmental Education," Peabody Journal of Education 51:5; October 1973.

2. Citizens Advisory Committee on Environmental Quality. Report to the President and to the President's Council on Environmental Quality. August 1969. p. 13.

3. Troost, Cornelius J., and Altman, Harold., editors. Environmental Education: A Sourcebook New York: John Wiley and Sons, Inc., 1972; and Welisch, Sophie, "Environmental Ecological Education." The Educational Forum 38:153-155, 1974.

4. Ris, Thomas F., editor. Energy and Man's Environment, Office of Environmental Programs, Superintendent of Public Instruction, Washington State. 1973. p. iv.

5. Ibid. p. 1.

The "U.S. Energy Crisis" curricular unit on energy provides a realistic framework for the study of the problems involved in the energy crisis today. Using a computer simulation, students have the opportunity to experiment with many energy-use patterns and different pollution standards and to see the effects not only on energy resources but on economic conditions and general public satisfaction.

Exercises of all types may be designed around the ENERGY simulation, allowing students to become actively involved in trying to solve the energy crisis and confronting the effects of different measures. Students may also analyze the mathematical model underlying ENERGY, thereby deepening their understanding of the interrelationships among energy issues and of the limits of the mathematical relationships used in this model.

Description of the ENERGY Program and Model

The ENERGY program simulates the effects of changes in 10 energy-use factors on overall energy supply and demand as well as on pollution levels, economic well-being, and general satisfaction. The factors which students may change are

Industry	1. Production Level
	2. Pollution Standards
Utilities	3. Production Level
	4. Pollution Standards
Transportation	5. Pollution Standards
	6. Mass Transportation
	7. Automobile Efficiency
Home	8. Appliances
	9. Home Heating and Lighting
Business and Schools	10. Heating and Lighting

Changes are entered as simple percentage increases or decreases in any or all factors each year, beginning with given conditions in 1975. Results are either listed or graphed for each year, with graphs showing the total supply and demand results for every year between 1975 and selected future years. Runs may be made for as many years as the user wishes.

The challenge of this program is for the user to introduce changes in order to keep demand for energy within the available energy supply and to reduce pollution while increasing economic well-being and general public satisfaction. For each factor not changed by the user each year, the program calculates

an automatic change, reflecting the trend of growing consumption. In addition, unexpected events are randomly introduced which increase the otherwise nonrenewable energy supply.

Directions for running ENERGY and a sample run are given in the Student Guide, pages 10-13.

The ENERGY model is limited to simulating effects on pollution, economic and satisfaction levels of changes in pollution standards and patterns of use of energy resources, assumed to be nonrenewable. It assumes that energy consumption is growing exponentially and that two things may extend the available supply of energy--changes in patterns of consumption and new energy sources being found. The model also assumes that changes in consumption patterns will directly influence the economic well-being of the nation and the general satisfaction of the people.

Basically, the relationships used by the ENERGY model are

INCREASES IN INDEPENDENT VARIABLES	RESULTING EFFECTS ON DEPENDENT VARIABLES				
	Energy Demand (ED)	Energy Resources (ES)	Pollution Level (PL)	Economic Well-Being (EWB)	General Satisfaction (GSL)
1. Industrial Production Level (IPL)	Inc.	Dec.	Inc.	Inc.	Inc.
2. Industrial Pollution Standards (IPS)	Inc.	Dec.	Dec.		
3. Utilities Production Level (UPL)	Inc.	Dec.	Inc.	Inc.	
4. Utilities Pollution Standards (UPS)	Inc.	Dec.	Dec.		
5. Transportation Pollution Standards (TPS)	Inc.	Dec.	Dec.		
6. Mass Transportation (TMT)	Dec.	Dec.			
7. Automobile Efficiency (TE)	Dec.	Dec.			
8. Appliances (A)	Inc.	Dec.			Inc.
9. Home Heating and Lighting (IHI)	Inc.	Dec.			Inc.
10. Schools and Business, Heating and Lighting (BSH)	Inc.	Dec.			Inc.

* Energy resources will constantly decrease unless all energy demand ceases. The only variable which can increase the resources is an "unexpected event" which is randomly introduced by the program itself.

The mathematical relationships are described on pages 17-19 of the Student Guide.

The data for the ENERGY simulation were drawn, for the most part, from the following three publications:

1. Hirst, Eric and John C. Moyers, "Efficiency of Energy Use in the United States," in Science, Vol. 179, March 30, 1973, pp. 1299-1304.

2. Patterns of Energy Consumption in the United States, prepared by Stanford Research Institute for the Office of Science and Technology. Washington: Government Printing Office, 1972.

3. A Time To Choose America's Energy Future. Final Report of the Energy Policy Project of the Ford Foundation. Cambridge, Mass.: Ballinger Publishing, 1974.

INTRODUCING STUDENTS TO THE SUBJECT AND UNIT

Introducing the Subject of the U.S. Energy Crisis

1. Students should read the background information on pages 1-7 of the Student Guide.

2. An excellent overview of the topic is provided in the 11-page article "Energy and the Environment," by John M. Fowler, in the December, 1972 issue of The Science Teacher.

3. Some of the activities suggested in the following sources will serve as good introductory activities:

   a. Energy and Man's Environment: Elementary through Secondary Interdisciplinary Activity Guide, available from Energy and Man's Environment, 2121 Fifth Avenue, Seattle, Washington 98121.

   b. "The Energy Question: Problems and Alternatives" in Intercom, No. 73, Teaching Toward Global Perspectives, available from Intercom, 218 East 18th Street, New York, New York 10003.

Preparing Students to Use the ENERGY Simulation.

1. Students should have a clear understanding of the limits of the ENERGY model before proceeding to use the simulation. The model's scope and the assumption on which it is based are described on pages 14-20 of the Student Guide. Students will discern additional assumptions as they work with the simulation and study the model.

   All students will need to understand the elementary relationships, between the energy-use factors and supply, demand, pollution, economic well-being and general satisfaction. The table on page 17 of the Student Guide identifies the general relationships in simple terms. The equations which determine the actual relationships used in the model are given on pages 17-19 of the Student Guide.

2. The model makes use of randomly introduced "news flashes" which announce the availability of new sources of energy and hence increase the otherwise nonrenewable energy resources. Students should understand that these events are included to simulate real-life possibilities which bear on energy availability.

3. Students should understand the procedures for running the ENERGY program, as outlined on pages 10-13 of the Student Guide.

12

STUDENT GUIDE EXERCISES: SUGGESTIONS AND ANSWERS

Suggestions

Exercise 1. All three parts of this exercise are sufficiently easy that all students may be expected to complete them. Scheduling of terminal time (about 10 minutes for each run) will require care to insure that all students can complete the exercise. In cases where time is limited, or other difficulties exist, parts (b) and (c) can be made optional for interested or advanced students. You may want students to submit their analyses of each run in a short written report. Comparisons of runs and student comments should be material for lively class discussion.

Exercise 2. The set of questions in this exercise will serve as a good overview of the ENERGY model's limitations. Class discussion of answers may be referred back to pages 10-20 of the Student Guide. This set of questions forms the basis of Follow-Up Activities, which are discussed in the next section of the Teacher's Guide.

Exercises 3. & 4. Average to advanced students may enjoy tackling this energy problem from these special angles. These exercises may be done individually or by committees of three to four students, who might compete for "election" on the basis of their 1975-1980 records and their new proposals.

Answers to Exercises

1. Runs and analyses will vary. Student grasp of the principle relationships in the ENERGY model should be reflected in improved strategies for runs (b) and (c).

2. a. 10

   b. No changes by student can increase energy supply, due to the models use of specifically nonrenewable energy sources (see pages 17 and 19 of the Student Guide).

   c. Answers will depend on strategies used and incidences of news flashes.

   d. "News flash" events increase energy supply by 2-10% (see page 20 of the Student Guide).

e. No. As the table on page 17 of the Student Guide shows, changes in factors 8, 9 and 10 do not cause changes in economic well-being. Some runs, however, may obscure this fact; hence, students should be referred to the Student Guide for clarification.

f. See again the table on page 17. Factors 1, 8, 9 and 10 alone effect general satisfaction, though runs may not show this effect clearly.

g. No consistent relationship between pollution level and general satisfaction has been included in the simulation. (See table, page 17.)

3. & 4. Runs will vary as will "Reports to the People."

FOLLOW-UP ACTIVITIES

I. The Limits and Scope of the ENERGY Model

1. On the basis of students experience with the ENERGY program and their answers to Exercise 2, the limits and biases of the ENERGY model may be discussed profitably by the class.

    a. As a preparation for class discussion, students may be asked to look over their answers to Exercise 2 and to note for each part how "realistic" they think the ENERGY model is.

    b. Suggestions may be gathered for changing the mathematical relationships in the model to be more realistic and advanced students with programming skills may be assigned the job of writing a revised program of their own. Further exercises could then be done with the student-written programs.

2. Special assignments can be made for researching independent variables excluded from the ENERGY model (such as population growth) and working out their effects within the ENERGY model. Advanced students may wish to create a new version of the program adding new variables.

II. Research Into Energy Sources and Issues

Any of the following will be interesting topics for students to research in order to broaden their understanding of the present crisis concerning energy resources, uses, and other issues. Brief class presentations can provide a focus for research and a way to maximize the exposure to information.

1. Present status and future prospects for new energy sources such as new fossil fuel deposits, geothermic energy and solar energy.

2. Current status of natural gas, oil, and coal resources and estimated future levels.

3. Energy use in any of the five societal areas in terms of consumption, waste, pollution and other factors.

4. Energy production and consumption in Western nations compared to that in Asian, South American or African nations.

5. The politics of oil.

6. Local ecological concerns related to energy production and consumption, such as land degradation or air pollution.

7. How pollution control devices work.

III. Research Into Ways and Means

An eye-opening experience for students will be to research the stages which measures for improving the energy situation must go through before measures can be implemented. Some specific areas for investigation might be:

1. Setting industrial pollution standards.

2. Setting automobile pollution control standards.

3. Establishing "energy-use" labelling laws for appliance manufacturers.

4. Establishing building standards to reduce energy waste.

5. Establishing legislation for or against nuclear power sources.

ANNOTATED SOURCE LIST[1]

Books and Articles

1. Citizens' Advisory Committee on Environmental Quality. Citizen Action Guide to Energy Conservation. Washington: Government Printing Office, 1973. 64 pp. $1.75

   Contains a factual account of the energy crisis, including practical tips for energy conservation. Offers suggestions on organizing for energy conservation efforts. List of selected references. Glossary of energy terms.

2. Freeman, S. David. "The Energy Crisis, What Makes It So Complex," Vital Issues, November 1973, p.6.

   The material was prepared primarily for upper elementary students.

3. Kraft, R. Wayne. "The World's Energy and Teilhard's Vision," America, December 15, 1973, pp. 457-460.

   Intriguing and scholarly examination of our present energy predicament and the concept of energy in the context of the universal human dilemma.

4. Landsberg, H. H., and S. H. Schurr. Energy in the United States: Sources, Uses and Policy Issues. New York: Random House, 1967.

   Intended to be popular. Non-Technical. A very coherent source on the pre-crisis energy picture. It is essentially a non-technical summary of two earlier reports on Resources for the Future Studies: Energy in the American Economy, 1850-1975 by Schurr and Nelschert (1960) and Resources in America's Future by Landsberg, Fischman, and Fisher (1963).

5. Large, David B., ed. Hidden Wastes: Potentials for Energy Conservation. Washington: Conservation Foundation, 1973.

   Analyzes methods for reducing energy waste.

---

1. Taken primarily from New York State United Teachers, Inc., "Energy Crisis Bibliography for Teacher Resources, available from NYSUT, 80 Wolf Road, Albany, New York 12205.

6. Scientific American, Energy and Power. San Francisco: W. H. Freeman and Co., 1971.

Eleven articles dealing with energy. Popular treatment with data presented in graphs and charts.

(Several other books are listed in the Student Guide.)

U.S. Government Publications

1. Office of Emergency Preparedness. The Potential for Energy Conservation: A Staff Study. Washington: Government Printing Office, 1972.

Energy Industry Sources

1. Public Affairs Department, Amoco Oil Company, Executive Park Drive, N.E., Atlanta, Georgia 30329. Ask for "The Energy Shortage: Our Side of the Story."

2. Shell Information Service, Shell Oil Company, P.O. Box 2463, Houston, Texas 77001. Information and teaching aids on oil and related subjects.

Audio-Visual Materials

1. "Energy and the Earth" Series, Earth: The Early Years, Part I (11 minutes, 56 frames, sd., solor, cassette); Earth: The Years of Decision, Part II (17 1/2 minutes, 69 frames, sd., color, cassette). Lyceum Productions, Inc., P. O. Box 1226, Laguna Beach, California 92652.

Views the earth from the origins of the solar system through its geological and biological development. Develops a fundamental understanding of our planet's energy resources and how they have been and are being expended. Environmental effects are carefully considered.

2. Energy: A Dialogue (12 audio tapes). Produced by the American Association for the Advancement of Science, Department N, 1515 Massachusetts Avenue, N.W., Washington, D.C. 20005. Edited by Norman Metzger.

3. "Environmental Crisis" Series. AAHPER/Publication Sales, NEA Center, 1201 Sixteenth Street, N.W., Washington, D.C. 20036.

A series of color filmstrips dealing with a variety of environmental topics. Open-ended, problem-oriented inter-disciplinary presentations for secondary school health and science courses and for professional preparation programs.

Set #1 Ecological priorities, pollution, animal wastes and population. Five 35 mm filmstrips, five LP records, and script.

Set #2 Industrial wastes, the energy crisis, consumer and community action, occupational health hazards, and economic considerations. Five 35 mm filmstrips, five LP records and script. Cassettes are also available.

Set #3 Industrial water pollution, the urban environment, the energy crisis: nuclear power, endangered species, and health effects of water pollution. Five 35 mm filmstrips, five LP records, and script. Cassettes are available.

Teaching Materials

1. "Energy Dialogue." Write to James Aldrich, Education Department, Conservation Foundation, 1717 Massachusetts Avenue, N.W., Washington, D.C. 20037.

   A simulation to help teachers and students "get into the energy crisis," is based on presentations and discussions by high school students, teachers, and environmentalists at a meeting this summer in New England.

2. "Snoopy Kit," free from Odom Fanning, Department of the Interior (South), Office of Energy Conservation, Room 465, 18th and East Streets, N.W., Washington, D.C. 20240.

   Features the famous "Peanuts" dog, includes various pamphlets and brochures on how to conserve energy.

PROGRAM LISTING

```
30  REM  AN ENVIRONMENTAL COMPUTER SIMULATION DEVELOPED BY
40  REM  THE NORTHWEST REGIONAL EDUCATIONAL LABORATORY--PORTLAND, OR
50  REM  PROGRAMMED IN 'BASIC' FOR THE HP 2000F
60  REM  WRITTEN BY JOHN R. LYNCH, SR.
70  REM  APRIL, 1974
80  REM  MODEL CONCEPTUALIZATION BY DAN KLASSEN AND DICK LYNCH
90  REM  UPDATED IN SEPTEMBER OF 1975 BY MICHAEL D. HISCOX-NWREL
100 PRINT
110 PRINT
120 PRINT
130 DIM A$[35],M[17],G[17],H[17]
140 PRINT
150 PRINT
160 PRINT TAB(12);"EEEEE   N   N   EEEEE   RRRRR   GGGGG   Y   Y"
170 PRINT TAB(12);"E       NN  N   E       R   R   G   G   Y   Y"
180 PRINT TAB(12);"E       N N N   E       R   R   G        Y Y"
190 PRINT TAB(12);"EEEE    N N N   EEEE    RRRRR   G        Y Y"
200 PRINT TAB(12);"E       N N N   E       R R     G GGG     Y"
210 PRINT TAB(12);"E       N  NN   E       R  R    G   G     Y"
220 PRINT TAB(12);"EEEEE   N   N   EEEEE   R   R   GGGGG     Y"
230 PRINT
240 PRINT
250 PRINT "YOU MAY RECORD RESULTS OF EACH RUN ON"
260 PRINT "THE BACK OF THE INSTRUCTION SHEET."
270 PRINT
280 PRINT
290 PRINT
300 PRINT
310 PRINT "ENERGY IS A SIMULATION OF THE ENERGY CRISIS.  YOU CAN TRY"
320 PRINT "TO SOLVE THE ENERGY PROBLEM BY CHANGING THE RATE AT WHICH"
330 PRINT "ENERGY IS USED.  YOUR GOAL IS TO KEEP THE DEMAND FOR ENERGY"
340 PRINT "WITHIN THE AVAILABLE SUPPLY.  AT THE SAME TIME YOU SHOULD"
350 PRINT "TRY TO"
360 PRINT "      > REDUCE THE LEVEL OF POLLUTION"
370 PRINT "      > INCREASE THE ECONOMIC CONDITION OF THE NATION"
380 PRINT "      > INCREASE THE GENERAL SATISFACTION OF THE PEOPLE"
390 PRINT
400 PRINT "YOU CAN CONTROL THE RATE AT WHICH ENERGY IS USED BY CHANGING"
410 PRINT "FACTORS IN TEN DIFFERENT AREAS.  EACH FACTOR AFFECTS THE"
420 PRINT "TOTAL AMOUNT OF ENERGY USED."
430 LET R=0
440 FOR J=1 TO 17
450 IF J>7 THEN 470
460 Z[J]=0
470 G[J]=130
480 H[J]=0
490 NEXT J
500 I=0
510 P8=74
520 Y=1974
530 RESTORE
540 FOR J=1 TO 10
```

```
550  S[J]=.386377
560  READ A[J]
570  E[J]=S[J]*A[J]
580  NEXT J
590  FOR J=1 TO 15
600  READ M[J]
610  NEXT J
620  FOR J=1 TO 10
630  READ P[J]
640  P[J]=P[J]*100
650  NEXT J
660  READ T9
670  T8=T9
680  DATA 1.2,.6,1.8,.4,1.5,1.6,1.4,1.8,1,1.2,27,20,22,14,5,3,1,2.5,1,1
690  DATA .6,.8,.5,2.5,2.5,.04,.02,.04,.02,.02,.04,.06,.06,.04,.02,80
700  R[1]=M[1]/2*M[6]*E[1]
710  R[2]=M[1]/2*M[7]*E[2]
720  R[3]=M[2]/2*M[8]*E[3]
730  R[4]=M[2]/2*M[9]*E[4]
740  R[5]=M[3]/3*M[10]*E[5]
750  R[6]=M[3]/3*M[11]*E[6]
760  R[7]=M[3]/3*M[12]*E[7]
770  R[8]=M[4]/2*M[13]*E[8]
780  R[9]=M[4]/2*M[14]*E[9]
790  R[10]=M[5]*M[15]*E[10]
800  FOR J=1 TO 10
810  R=R+R[J]
820  NEXT J
830  C[1]=R[1]+R[2]
840  C[2]=R[3]+R[4]
849  V=20
850  C[3]=V+R[5]-R[6]-R[7]
860  C[4]=R[8]+R[9]
870  C[5]=R[10]
880  J=70/905.85
890  C[8]=(E[1]+E[8]+E[9]+E[10])/J
900  C[7]=(E[1]+E[3])/J
910  C[6]=20-(E[2]+E[4]+E[5])/3+.1*(E[1]+E[3])
920  FOR J=1 TO 8
930  C[J]=INT(C[J]*100+.5)/100
940  NEXT J
950  PRINT
960  G[1]=T9
970  H[1]=P8
980  PRINT
990  PRINT
1000  IF I<1 THEN 1870
1001  V=V*1.07
1010  PRINT "WANT RESULTS GRAPHED";
1020  INPUT A$
1030  PRINT
1040  PRINT
1050  PRINT
1060  PRINT
```

```
1070  I=I+1
1080  P8=0
1090  U=35
1100  L=55
1110  U3=(U-L)/14
1120  U4=0
1130  IF I<2 THEN 1150
1140  PRINT "END OF YEAR -";Y
1141  PRINT
1142  PRINT
1150  FOR J<U TO L STEP -U3
1160  U4=U4+1
1170  IF A$[1,1]="N" THEN 1380
1180  IF INT(J+.5)=135 THEN 1220
1190  IF INT(J+.5)=95 THEN 1220
1200  IF INT(J+.5)=55 THEN 1220
1210  GOTO 1230
1220  PRINT INT(J+.5);
1230  PRINT TAB(8);"I";
1240  FOR K=1 TO I
1250  IF G[K] >= J+U3 THEN 1340
1260  IF G[K]<J THEN 1340
1270  PRINT TAB(7+2*K);
1280  IF H[K] >= J+U3 THEN 1320
1290  IF H[K]<J THEN 1320
1300  PRINT "E";
1310  GOTO 1370
1320  PRINT "S";
1330  GOTO 1370
1340  IF H[K] >= J+U3 THEN 1370
1350  IF H[K]<J THEN 1370
1360  PRINT TAB(7+2*K);"D";
1370  NEXT K
1380  IF U4 <> 1 THEN 1420
1390  IF A$[1,1]="Y" THEN 1410
1400  PRINT "ENERGY SUPPLY -";INT(G[I]+.5);
1410  PRINT TAB(51);"POLLUTION LEVEL";
1420  IF U4 <> 2 THEN 1440
1430  PRINT TAB(56);INT(C[6]*100+.5)/100;
1440  IF U4 <> 5 THEN 1480
1450  IF A$[1,1]="Y" THEN 1470
1460  PRINT "ENERGY DEMAND -";INT(H[I]+.5);
1470  PRINT TAB(49);"ECONOMIC WELL-BEING";
1480  IF U4 <> 6 THEN 1500
1490  PRINT TAB(56);INT(C[7]*100+.5)/100;
1500  IF U4 <> 9 THEN 1580
1510  IF A$[1,1]="Y" THEN 1570
1520  IF INT(G[I]-H[I]+.5) <= 0 THEN 1550
1530  PRINT "SURPLUS ENERGY -";
1540  GOTO 1560
1550  PRINT "ENERGY DEFICIT -";
1560  PRINT ABS(INT(G[I]-H[I]+.5));
1570  PRINT TAB(49);"GENERAL SATISFACTION";
1580  IF U4 <> 10 THEN 1600
```

```
1590 PRINT TAB(56);INT(C[8]*100+.5)/100;
1600 IF U4 <> 13 THEN 1630
1610 IF A$[1,1]="N" THEN 1630
1620 PRINT TAB(48);"S = SUPPLY OF ENERGY";
1630 IF U4 <> 14 THEN 1660
1640 IF A$[1,1]="N" THEN 1660
1650 PRINT TAB(48);"D = ENERGY DEMAND";
1660 IF U4 <> 15 THEN 1690
1670 IF A$[1,1]="N" THEN 1700
1680 PRINT TAB(48);"E = DEMAND EQUALS SUPPLY";
1690 PRINT
1700 NEXT J
1710 IF A$[1,1]="N" THEN 1740
1720 PRINT TAB(8);"I- - - - - - - - - - - - - - - - - - -"
1730 PRINT TAB(8);"1975 76  78  80  82  84  86  88  90"
1740 PRINT
1750 Y=Y+1
1760 IF I<17 THEN 1780
1770 STOP
1780 IF I<2 THEN 2470
1790 PRINT "DO YOU WANT A LIST OF CURRENT ENERGY DEMANDS? ";
1800 INPUT A$
1810 PRINT
1820 IF A$[1,1] <> "S" THEN 1840
1830 STOP
1840 IF A$[1,1]="N" THEN 2470
1850 IF A$[1,1] <> "Y" THEN 1790
1860 REM  THIS IS FACTOR
1870 IF I >= 1 THEN 1910
1880 PRINT "START OF YEAR - 1975"
1890 PRINT
1900 GOTO 1950
1910 PRINT
1920 PRINT
1930 PRINT
1940 PRINT
1950 PRINT TAB(5);
1960 PRINT "CURRENT ENERGY DEMAND:";TAB(49);"FACTORS IN AREAS"
1970 PRINT TAB(5);"BTU TIMES 10 TO THE 15TH";TAB(43);"ID#   YOU CAN CHA
1980 FOR J=1 TO 5
1990 PRINT
2000 PRINT TAB(6);INT(C[J]+.5);TAB(18);
2010 IF J <> 1 THEN 2030
2020 PRINT "INDUSTRY";
2030 IF J <> 2 THEN 2050
2040 PRINT "UTILITIES";
2050 IF J <> 3 THEN 2070
2060 PRINT "TRANSPORTATION";
2070 IF J <> 4 THEN 2090
2080 PRINT "HOMES";
2090 IF J <> 5 THEN 2110
2100 PRINT "BUSINESS AND SCHOOLS";
2110 PRINT TAB(44);
2120 IF J <> 1 THEN 2170
```

```
2130  PRINT "1     ";
2140  GOSUB 2350
2150  PRINT TAB(44);"2     ";
2160  GOSUB 2370
2170  IF J <> 2 THEN 2220
2180  PRINT "3     ";
2190  GOSUB 2350
2200  PRINT TAB(44);"4     ";
2210  GOSUB 2370
2220  IF J <> 3 THEN 2270
2230  PRINT "5     ";
2240  GOSUB 2370
2250  PRINT TAB(44);"6     MASS TRANSPORTATION"
2260  PRINT TAB(44);"7     AUTOMOBILE EFFICIENCY"
2270  IF J <> 4 THEN 2310
2280  PRINT "8     APPLIANCES              "
2290  PRINT TAB(44);"9     ";
2300  GOSUB 2390
2310  IF J <> 5 THEN 2410
2320  PRINT "10    ";
2330  GOSUB 2390
2340  GOTO 2410
2350  PRINT " PRODUCTION LEVEL"
2360  RETURN
2370  PRINT " POLLUTION STANDARDS"
2380  RETURN
2390  PRINT " HEATING / LIGHTING"
2400  RETURN
2410  NEXT J
2420  PRINT
2430  IF I<1 THEN 1010
2440  IF I<17 THEN 2470
2450  A$="YES"
2460  GOTO 1030
2470  P8=0
2480  PRINT
2490  PRINT
2500  PRINT "DO YOU WANT TO CHANGE ANY FACTORS? ";
2510  INPUT A$
2520  PRINT
2530  IF A$[1,1] <> "S" THEN 2550
2540  STOP
2550  IF A$[1,1]="N" THEN 3090
2560  IF A$[1,1] <> "Y" THEN 2480
2570  FOR J=1 TO 10
2580  K[J]=-.0694
2590  NEXT J
2600  FOR J=1 TO 10
2610  S[J]=E[J]/A[J]
2620  NEXT J
2630  PRINT
2640  PRINT "TO MAKE A CHANGE, ENTER THE FACTOR ID NUMBER, A COMMA"
2650  PRINT "AND THE PERCENT CHANGE."
2660  PRINT
```

```
2670 PRINT "FOR EXAMPLE, TO REDUCE THE USE OF APPLIANCES IN THE HOME"
2680 PRINT "BY 25% YOU ENTER 8,-25"
2690 PRINT
2700 PRINT "WHEN YOU ARE DONE ENTERING CHANGES, ENTER 0,0 "
2710 INPUT I5,I6
2720 IF I5=0 THEN 2840
2730 I5=INT(ABS(I5))
2740 IF I5 <= 10 THEN 2820
2750 PRINT "ENTER ONLY ID FROM 1 TO 10 "
2760 GOTO 2710
2770 IF ABS(I6) <= 75 THEN 2820
2780 IF I<0 THEN 2810
2790 I6=75
2800 GOTO 2820
2810 I6=-75
2820 K[I5]=I6
2830 GOTO 2710
2840 FOR J=1 TO 10
2850 IF K[J] <> -.0694 THEN 2880
2860 E[J]=E[J]*(1+P[J]/100)
2870 GOTO 2900
2880 S[J]=S[J]*(1+K[J]/100)
2890 E[J]=A[J]*S[J]
2900 NEXT J
2910 PRINT
2920 T7=T9
2930 GOSUB 3150
2940 IF R <= T9 THEN 3440
2950 FOR J<1 TO 10
2952 K[J]=0
2960 IF K[J]=-.0694 THEN 3020
2970 R=R-R[J]
2980 T7=T7-R[J]
2990 IF T7>0 THEN 3020
3000 GOSUB 3150
3010 GOTO 3300
3020 NEXT J
3030 T7=(R-T7)/R
3040 FOR J<1 TO 10
3050 IF K[J] <> -.0694 THEN 3070
3070 NEXT J
3080 GOTO 3340
3090 FOR J=1 TO 10
3100 E[J]=E[J]*(1+P[J]/100)
3110 NEXT J
3120 T7=T9
3130 GOSUB 3150
3140 GOTO 2940
3150 R[1]=M[1]/2*M[6]*E[1]
3160 R[2]=M[1]/2*M[7]*E[2]
3170 R[3]=M[2]/2*M[8]*E[3]
3180 R[4]=M[2]/2*M[9]*E[4]
3190 R[5]=M[3]/3*M[10]*E[5]
3200 R[6]=M[3]/3*M[11]*E[6]
```

```
3210  R[7]=M[3]/3*M[12]*E[7]
3220  R[8]=M[4]/2*M[13]*E[8]
3230  R[9]=M[ ]/2*M[14]*E[9]
3240  R[10]=M[5]*M[15]*E[10]
3250  R=0
2260  FOR J=1 TO 10
3270  R=R+R[J]
3280  NEXT J
3290  RETURN
3300  R=(R-T9)/R
3310  FOR J=1 TO 10
3320  R[J]=R[J]*(1-R)
3330  NEXT J
3340  E[1]=R[1]*2/M[1]/M[6]
3350  E[2]=R[2]*2/M[1]/M[7]
3360  E[3]=R[3]*2/M[2]/M[8]
3370  E[4]=R[4]*2/M[2]/M[9]
3380  E[5]=R[5]*3/M[3]/M[10]
3390  E[6]=R[6]*3/M[3]/M[11]
3400  E[ ]=R[7]*3/M[3]/M[12]
3410  E[8]=R[8]*2/M[4]/M[13]
3420  E[9]=R[9]*2/M[4]/M[14]
3430  E[10]=R[10]/M[5]/M[15]
3440  C[1]=R[1]+R[2]
3450  C[2]=R[3]+R[4]
3460  C[3]=V+R[5]-R[6]-R[7]
3470  C[4]=R[8]+R[9]
3480  C[5]=R[10]
3490  J=70/905.85
3500  C[8]=(E[1]+E[8]+E[9]+E[10])/J
3510  C[7]=(E[1]+E[3])/J
3520  C[6]=20-(E[2]+E[4]+E[5])/3+.1*(E[1]+E[3])
3530  FOR J=1 TO 8
3540  C[J]=INT(C[J]*100+.5)/100
3550  IF J>5 THEN 3570
3560  P8=C[J]+P8
3570  NEXT J
3580  G[I+1]=T9
3590  H[I+1]=P8
3600  T9=T9-P8
3610  T8=T8*.99
3620  T9=T9+T8
3630  PRINT
3640  IF I<3 THEN 990
3650  X=INT(RND(1)*9+1.5)
3660  IF X=2 THEN 3700
3670  IF X=4 THEN 3700
3680  IF X=7 THEN 3700
3690  IF X <> 9 THEN 990
3700  IF Z[6] <> 0 THEN 990
3710  PRINT
3720  PRINT
3730  GOSUB 3750
```

```
3740 GOTO 3790
3750 FOR J=1 TO 4
3760 PRINT "* ";
3770 NEXT J
3780 RETURN
3790 PRINT " NEWS FLASH ";
3800 GOSUB 3750
3810 PRINT
3820 PRINT
3830 X8=INT(RND(1)*8+2.5)
3840 PRINT
3850 PRINT "BULLETIN--";
3860 X=INT(RND(1)*16+.5)
3870 IF X<1 THEN 3860
3880 IF X>15 THEN 3860
3890 FOR J=1 TO 6
3900 IF X=Z[J] THEN 3860
3910 IF Z[J] <> 0 THEN 3940
3920 Z[J]=X
3930 GOTO 3950
3940 NEXT J
3950 IF X=5 THEN 4110
3960 IF X=2 THEN 4150
3970 IF X=10 THEN 4190
3980 IF X=3 THEN 4220
3990 IF X=9 THEN 4250
4000 IF X=14 THEN 4290
4010 IF X=7 THEN 4320
4020 IF X=12 THEN 4350
4030 IF X=1 THEN 4390
4040 IF X=15 THEN 4430
4050 IF X=8 THEN 4460
4060 IF X=13 THEN 4520
4070 IF X=4 THEN 4540
4080 IF X=6 THEN 4570
4090 IF X=11 THEN 4600
4100 GOTO 3860
4110 X5=1
4120 PRINT "DEVELOPMENT OF METHANE"
4130 PRINT "GENERATION TECHNIQUES"
4140 GOTO 4630
4150 X5=2
4160 PRINT "THE USE OF NUCLEAR EXPLOSIVES IN THE"
4170 PRINT "PRODUCTION OF OIL AND NATURAL GAS"
4180 GOTO 4630
4190 X5=3
4200 PRINT "THE USE OF SOLAR POWERED FUEL CELLS"
4210 GOTO 4630
4220 X5=4
4230 PRINT "THE DEVELOPMENT OF WIND POWER"
4240 GOTO 4630
4250 X5=5
4260 PRINT "THE REOPENING OF PREVIOUSLY"
4270 PRINT "UNECONOMICAL COAL MINES"
```

```
4280  GOTO 4630
4290  X5=6
4300  PRINT "FAST BREEDER REACTOR TECHNOLOGY"
4310  GOTO 4630
4320  X5=7
4330  PRINT "LARGE SCALE SHALE OIL RECOVERY"
4340  GOTO 4630
4350  X5=8
4360  PRINT "ADAPTING POWER PLANTS TO"
4370  PRINT "BURN LIGNITE COAL"
4380  GOTO 4630
4390  X5=9
4400  PRINT "THE INCREASED USE OF SOLAR"
4410  PRINT "HOME HEATING UNITS"
4420  GOTO 4630
4430  X5=10
4440  PRINT "WIDESPREAD BURNING OF REFUSE FOR POWER"
4450  GOTO 4630
4460  X5=11
4470  PRINT "INCREASED USE OF"
4480  PRINT "GEOTHERMAL ENERGY SOURCES"
4490  GOTO 4630
4500  X5=12
4510  PRINT "COMPLETION OF SEVERAL TIDAL"
4520  PRINT "POWDR GENERATION PLANTS"
4530  GOTO 4630
4540  X5=13
4550  PRINT "CONTROLLED THERMONUCLEAR FUSION POWER"
4560  GOTO 4630
4570  X5=14
4580  PRINT "LASER POWER TRANSMISSION"
4590  GOTO 4630
4600  X5=15
4610  PRINT "RELAY OB SOLAR ENERGY VIA"
4620  PRINT "SATELLITE COLLECTORS"
4630  PRINT "HAS BEEN SUCCESSFUL IN"
4640  PRINT "INCREASING THE YEARLY ENERGY SUPPLY ";X8;"%"
4650  T9=T9-T8
4660  T8=T8*(1+X8/100)
4670  T9=T9+T8
4680  PRINT
4690  PRINT
4700  PRINT
4710  G[I+1]=T9
4720  PRINT
4730  PRINT
4740  GOTO 990
4750  END
```